Dice Probabilities 02-26-16

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Purpose

I develop some experience for using the new *Mathematica* probability capabilities by using them to solve problems involving dice. Some of these problems are classic. Some I found on the web — *A Collection of Dice Problems (v 1-2-2010, Matthew M. Conroy.* Of course, there are many ways to approach and solve these problems. I tend to use approaches that make use of *Mathematica* capabili-ties.

Probability sum of dice will be less than certain amount

What is the probability that the total of the face values of two dice will be 6 or less?

Model the die throw with a uniform distribution

```
Clear[FairDie];
FairDie = DiscreteUniformDistribution[{1, 6}];
```

The required probability is

```
Probability[d1 + d2 ≤ 6, {d1 ≈ FairDie, d2 ≈ FairDie}]
```

$$\frac{5}{12}$$

The general result is also available

copyright © N T Gladd 2016

```
Probability[d1 + d2 ≤ n, {d1 ≈ FairDie, d2 ≈ FairDie}]
```

$$\begin{cases} \frac{1}{36} & 2 \le n < 3 \\ \frac{1}{12} & 3 \le n < 4 \\ \frac{1}{6} & 4 \le n < 5 \\ \frac{5}{18} & 5 \le n < 6 \\ \frac{5}{12} & 6 \le n < 7 \\ \frac{7}{12} & 7 \le n < 8 \\ \frac{13}{18} & 8 \le n < 9 \\ \frac{5}{6} & 9 \le n < 10 \\ \frac{11}{12} & 10 \le n < 11 \\ \frac{35}{36} & 11 \le n < 12 \\ 1 & n \ge 12 \\ 0 & \text{True} \end{cases}$$

I simulate so as to have independent confirmation of results.

```
Module[{nTimes = 1000, f, theory, theoryList, lab},
 theory = Probability[d1 + d2 ≤ n, {d1 ≈ FairDie, d2 ≈ FairDie}];
 theoryList = Transpose[{Range[2, 12], theory⟦1⟧⟦All, 1⟧}];

 f = EmpiricalDistribution[
   Table[RandomVariate[FairDie] + RandomVariate[FairDie], {nTimes}]];
 lab = Stl@StringForm[
    "Comparison of theoretical CDF (red) with simulated CDF\n simulation size = ``",
    nTimes];
 DiscretePlot[CDF[f, n], {n, 1, 12}, ExtentSize → Right,
  Epilog → {PointSize[0.015], Red, Point /@ theoryList},
  PlotLabel → lab, PlotStyle → Black]]
```



copyright © N T Gladd 2016

Expected time for a six to appear

Suppose the target number during a sequence of dice rolls is 6. On the first roll, 6 will occur with probability $\frac{1}{6}$ and some other number with probability $\frac{5}{6}$. The probability that 6 will occur on a particular roll is (for p = 1/6 and q = 5/6 = 1 - p)

1 p
2 q p
3 q^2 p
...
k q^{k-1}p

This sequence forms a geometric distribution

```
Module[{p = 1/6, fDist, results},
 fDist = GeometricDistribution[p];
 results = Table[{i, Probability[x ≤ i, x ≈ fDist] // N}, {i, 0, 10}];
 LGrid[results, "CDF[first occurrence of 6]"]]
```

CDF[first occurrence of 6]

0	0.166667
1	0.305556
2	0.421296
3	0.517747
4	0.598122
5	0.665102
6	0.720918
7	0.767432
8	0.806193
9	0.838494
10	0.865412

Check against simulation

copyright © N T Gladd 2016

```
Module[{p = 1/6, nTrials = 1000,
  nRolls = 50, results, fFirstSix, theory, lab, FirstSix},
 FirstSix[n_] :=
  Position[Table[RandomVariate[FairDie], {n}], 6][[1, 1]];

 theory =
  Table[{i + 1, Probability[x ≤ i, x ≈ GeometricDistribution[p]] // N}, {i, 0, 10}];
 lab = Stl@StringForm["Probability a 6 will first appear on
     the n^th roll\nTheory(red) vs 1000 trial simulation"];
 results = Table[FirstSix[nRolls], {nTrials}];
 fFirstSix = EmpiricalDistribution[results];
 DiscretePlot[CDF[fFirstSix, x], {x, 0, 10}, PlotStyle → Black,
  ExtentSize → Right, AxesLabel → {Stl["nRolls"], Stl["ℙ"]},
  PlotLabel → lab, Epilog → {Red, PointSize[0.015], Point /@ theory}]]
```



What is the expected number of rolls before a six is thrown? What is the standard deviation about this number?

```
{Mean[GeometricDistribution[p]], StandardDeviation[GeometricDistribution[p]]}
```

$$\left\{-1+\frac{1}{p}, \frac{\sqrt{1-p}}{p}\right\}$$

```
{Mean[GeometricDistribution[1/6]],
  StandardDeviation[GeometricDistribution[1/6]]} // N
```

```
{5., 5.47723}
```

What is the probability a 6 will appear on the 4th roll?

```
Probability[x == 4, x ≈ GeometricDistribution[1/6]] // N
```

```
0.0803755
```

copyright © N T Gladd 2016

What is the probability a 6 will appear on or before the 4th roll?

```
Probability[x <= 4, x ≈ GeometricDistribution[1/6]] // N
```

```
0.598122
```

What is the probability no 6 will appear in the first 10 rolls?

```
Probability[x > 10, x ≈ GeometricDistribution[1/6]] // N
```

```
0.134588
```

How many other numbers are expected to be rolled before six appears twice?

The relevant distribution in this case is the negative binomial distribution. This is defined by

If n is a positive integer, NegativeBinomialDistribution$[n, p]$ gives the distribution of the number of failures in a sequence of trials with success probability p before n successes occur.

```
Module[{p = 1/6, lab},
 lab = Stl@StringForm[
    "CDF of number of other numbers appearing\nbefore 1, 2, 3 sixes appear"];
 DiscretePlot[{CDF[NegativeBinomialDistribution[1, p], x],
   CDF[NegativeBinomialDistribution[2, p], x],
   CDF[NegativeBinomialDistribution[3, p], x]}, {x, 0, 10},
  PlotLabel → lab, AxesLabel → {Stl["n"], Stl["ℙCDF"]}]]
```



The probability that 5 numbers will be rolled before 6 appears twice is

```
Probability[x <= 5, x ≈ NegativeBinomialDistribution[2, 1/6]] // N
```

```
0.330204
```

The expected number of “other numbers” before 2 sixes appear is

```
{Mean[NegativeBinomialDistribution[2, 1/6]],
  StandardDeviation[NegativeBinomialDistribution[2, 1/6]]} // N
```

```
{10., 7.74597}
```

Another useful quantity is the distribution of the number of 6s that appear in a sequence of n rolls

```
Module[{p = 1/6, nRolls = 30, fDist},
 fDist = BinomialDistribution[nRolls, p];
 DiscretePlot[PDF[fDist, x], {x, 0, 10}]]
```




```
Module[{p = 1/6, nRolls = 30, fDist},
 fDist = BinomialDistribution[nRolls, p];
 DiscretePlot[{PDF[BinomialDistribution[10, p], x],
   PDF[BinomialDistribution[20, p], x], PDF[BinomialDistribution[30, p], x]},
  {x, 0, 10}, Joined → True, AxesLabel → {Stl["n 6s"], Stl["ℙ"]},
  PlotLegends → {"10 rolls", "20 rolls", "30 rolls"}]]
```



copyright © N T Gladd 2016

Chevalier de Mere’s problem

Chevalier de Méré's Problem

A 17^{th} century gambler, the Chevalier de Méré, made it to history by turning to Blaise Pascal for an explanation of his unexpectd losses. Pascal combined his efforts with his friend Pierre de Fermat and the two of them laid out mathematical foundations for the theory of probability.

Gamblers in the 1717 France were used to bet on the event of getting at least one 1 (ace) in four rolls of a dice. As a more trying variation, two die were rolled 24 times with a bet on having at least one double ace. According to the reasoning of Chevalier de Méré, two aces in two rolls are 1/6 as likely as 1 ace in one roll. (Which is correct.) To compensate, de Méré thought, the two die should be rolled 6 times. And to achieve the probability of 1 ace in four rolls, the number of the rolls should be increased four fold - to 24. Thus reasoned Chevalier de Méré who expected a couple of aces to turn up in 24 double rolls with the frequency of an ace in 4 single rolls. However, he lost consistently.

What is the probability an ace (a “1”) will appear in 4 rolls of a die?

```
Probability[x <= 4, x ≈ GeometricDistribution[1 / 6]] // N

0.598122
```

What is the probability double aces (a “1”) will appear in 24 rolls of a die?

```
Probability[x <= 24, x ≈ GeometricDistribution[1 / 36]] // N

0.505532
```

The Chevalier was betting as if these two probabilities were the same — and losing. This is a seminal problem in probability. De Mere (Antoine Gombaud) turned to Blaise Pascal, who enlisted Pierre de Fermat and thus began the modern theory of probability (see Wikipedia for more).

copyright © N T Gladd 2016

```
Module[{lab},
 lab =
  Stl@StringForm["Number of rolls of 2 dice required\nto throw 1-1 with the same
     probability\nas throwing 1 in 4 rolls of one die"];
 DiscretePlot[Probability[x <= n, x ≈ GeometricDistribution[1/36]], {n, 20, 40},
  Epilog → {Red, Line[{{20, 0.5981224279835391`}, {40, 0.5981224279835391`}}]},
  PlotStyle → Black, PlotLabel → lab ]]
```



Pepys problem posed to Newton

7. What is the most probable: rolling at least one six with six dice, at least two sixes with twelve dice, or at least three sixes with eighteen dice? (This is an old problem, frequently connected with Isaac Newton.)

The history of this problem may be found in
http://galton.uchicago.edu/faculty/stigler/pubs/IsaacNewtonProb-final.html
Newton got the right answer but the logic of his explanation was wrong! This is the only known work by Newton involving probability.

```
Module[{p = 1/6, nRolls = 6},
 Probability[x ≥ 1, x ≈ BinomialDistribution[nRolls, p]] // N]

0.665102
```

```
Module[{p = 1/6, nRolls = 12},
 Probability[x ≥ 2, x ≈ BinomialDistribution[nRolls, p]] // N]

0.618667
```

copyright © N T Gladd 2016

```
Module[{p = 1/6, nRolls = 18},
 Probability[x ≥ 3, x ≈ BinomialDistribution[nRolls, p]] // N]
```

```
0.597346
```

I also write a simulation

```
Clear[RollnDice, MonteCarloDiceCountSixes];
RollnDice[n_] := Table[RandomInteger[{1, 6}], {n}];
MonteCarloDiceCountSixes[nDice_, nSixes_, nRolls_] :=
 Module[{rolls, counts, successes},
  (* roll nDice dice, nRolls times *)
  rolls = Table[RollnDice[nDice], {nRolls}];
  (* count how many 6s occur on each roll *)
  counts = (Count[#, 6]) & /@ rolls;
  (* mark success as 1 if more than nSixes occurred *)
  successes = (If[# ≥ nSixes, 1, 0]) & /@ counts;
  (* Take the average to get the Monte Carlo estimate *)
  Mean[successes] // N]
```

```
Module[{theory, sim, results, nRolls = 10000},
 theory = {0.665102`, 0.618667`, 0.597345`};
 sim = {MonteCarloDiceCountSixes[6, 1, nRolls],
   MonteCarloDiceCountSixes[12, 2, nRolls],
   MonteCarloDiceCountSixes[18, 3, nRolls]};
 results = Transpose[{{"at least 1 in 6 rolls",
     "at least 2 in 12 rolls", "at least 3 in 18 rolls"}, theory, sim}];
 PrependTo[results, {"", "theory", "MonteCarlo"}];
 LGrid[results, "Comparing theory with simulation"]]
```

Comparing theory with simulation

	theory	MonteCarlo
at least 1 in 6 rolls	0.665102	0.661
at least 2 in 12 rolls	0.618667	0.6215
at least 3 in 18 rolls	0.597345	0.5948

What is the probability of winning a craps hand?

Two dice are used in a craps game. The basic probability distribution can be represented by

```
crapsRoll = ProductDistribution[{DiscreteUniformDistribution[{1, 6}], 2}]
```

```
ProductDistribution[{DiscreteUniformDistribution[{1, 6}], 2}]
```

The probability of winning on the initial or “come out” roll (making 7 or 11)

copyright © N T Gladd 2016

```
ℙ["7-11"] = Probability[Or[D1 + D2 == 7 ⋁ D1 + D2 == 11], {D1, D2} ≈ crapsRoll]
```

$\frac{2}{9}$

The probability of losing (“crapping out” by making 2 or 3 or 12) on the “come out” roll

```
Probability[Or[D1 + D2 == 2 ⋁ D1 + D2 == 3 ⋁ D1 + D2 == 12], {D1, D2} ≈ crapsRoll]
```

$\frac{1}{9}$

If one of the numbers 4,5,6 or 8,9,10 are rolled, that number becomes the “point” and the player must then roll the “point” before rolling seven. The probability of rolling
4 or 10 is 3/36
5 or 9 is 4/36
6 or 8 is 5/36

```
{{Probability[D1 + D2 == 4, {D1, D2} ≈ crapsRoll],
  Probability[D1 + D2 == 10, {D1, D2} ≈ crapsRoll]},
 {Probability[D1 + D2 == 5, {D1, D2} ≈ crapsRoll],
  Probability[D1 + D2 == 9, {D1, D2} ≈ crapsRoll]},
 {Probability[D1 + D2 == 6, {D1, D2} ≈ crapsRoll],
  Probability[D1 + D2 == 8, {D1, D2} ≈ crapsRoll]}}
```

$\left\{\left\{\frac{1}{12}, \frac{1}{12}\right\}, \left\{\frac{1}{9}, \frac{1}{9}\right\}, \left\{\frac{5}{36}, \frac{5}{36}\right\}\right\}$

After establishing a point, each successive roll consists winning by rolling the point, losing by rolling 7, or continuing if some other number is rolled. The probability of winning on the n^{th} roll (where n ≥ 0)

$$\mathbb{P}(\text{point})\,\mathbb{P}(\text{not 7 or point})^n$$

The overall probability of rolling a point and then winning the point is

$$\mathbb{P}[\text{point}]\sum_{n=0}^{\infty}\mathbb{P}[\text{not 7 or point}]^n$$

For example, suppose the point is 4 (or 10), then ℙ[4] = 3/36, ℙ[not 4 and not 7] = 1 - 3/36 - 6/36 = 2/3. Then, condition on the original point roll

$$\left\{\mathbb{P}\text{point}[4] = \frac{3}{36}\times\sum_{n=0}^{\infty}\frac{3}{36}\left(1-\frac{3}{36}-\frac{6}{36}\right)^n,\ \mathbb{P}\text{point}[5] = \frac{4}{36}\times\sum_{n=0}^{\infty}\frac{4}{36}\left(1-\frac{4}{36}-\frac{6}{36}\right)^n,\ \mathbb{P}\text{point}[6] = \frac{5}{36}\times\sum_{n=0}^{\infty}\frac{5}{36}\left(1-\frac{5}{36}-\frac{6}{36}\right)^n\right\}$$

$\left\{\frac{1}{36}, \frac{2}{45}, \frac{25}{396}\right\}$

copyright © N T Gladd 2016

Using the symmetry between 4/10, 5/9, 6/8, the overall probability of winning at craps is

```
ℙ["win hand"] = ℙ["7-11"] + 2 ℙpoint[4] + 2 ℙpoint[5] + 2 ℙpoint[6]
```

$$\frac{244}{495}$$

or

```
ℙ["win hand"] // N

0.492929
```

The odds favor the house, not the player.

What is the expected number of rolls for all sides of a die to appear?

Derivation of the answer.

On the first roll, one of the six numbers must have appeared. There is probability 5/6 that a second number will appear on the next roll. But the expected number of rolls for an event with probability 5/6 to occur is 6/5. After the second number has appeared, the probability that the 3rd number will appear on the next roll is 4/6. The expected time for that event to occur is 6/4. Continuing this line of thought

```
ℙ = 1 + 6 / 5 + 6 / 4 + 6 / 3 + 6 / 2 + 6 / 1 // N

14.7
```

I first consider a simulation approach drawing on material from notebook *Some Dice Problems 2*

copyright © N T Gladd 2016

```
Module[{nTrials = 10 000, nMax = 100, results, 𝒟, theoryLine, lab},
 results = Table[FirstOccurrenceOfAllSixNumbers[nMax], {nTrials}];
 𝒟 = EmpiricalDistribution[results];
 lab = Stl@StringForm[
    "𝔼[roll on which all sides have appeared] = ``\ntheory = 14.7", N[Mean[𝒟]] ];
 theoryLine = {Red, Line[{{14.7, 0}, {14.7, 100}}] };
 DiscretePlot[PDF[𝒟, x], {x, 1, 50},
  PlotStyle → Black, PlotLabel → lab, Epilog → theoryLine]]
```




```
Clear[FirstOccurrenceOfAllSixNumbers];
FirstOccurrenceOfAllSixNumbers[nMax_] :=
 Module[{listOfRolls},
  listOfRolls = RandomVariate[FairDie, nMax];
  i = 5;
  While[Length[Union[listOfRolls〚1 ;; i〛]] < 6,
   i = i + 1];
  i]
```

What is the distribution of the highest value resulting from a throw of n dice?

Derivation: Consider the case n = 3.
Suppose the highest number rolled is 6. The probability that all numbers rolled are less than k is $k^3/6^3$. However, some of the die do not have a k, but a number between 1 and k -1. The probability associated with such dice is $(k-1)^3/6^3$. So the probability that k is the highest number is
$k^3/6^3 - (k-1)^3/6^3$

```
Clear[HighestNumberDistribution];
HighestNumberDistribution[k_] :=
```

$$\frac{k^3}{6^3} - \frac{(k-1)^3}{6^3}$$

copyright © N T Gladd 2016

```
Module[{nDice = 3, nTrials = 1000, results, theoryLine, lab, 𝒟},
 results = Table[LargestNumber[nDice], {nTrials}];
 𝒟 = EmpiricalDistribution[results];
 lab =
  Stl@StringForm["Distribution of highest number from `` dice\nBlack is theory,
      red is simulation of `` trials", nDice, nTrials];
 theoryLine = {Red, Line[{{14.7, 0}, {14.7, 100}}] };
 DiscretePlot[{HighestNumberDistribution[x] , PDF[𝒟, x]},
  {x, 1, 6}, PlotLabel → lab, PlotStyle → {Black, Red},
  AxesLabel → {Stl["highest number"], Stl["ℙ"]}]]
```



Simulation

```
Clear[LargestNumber];
LargestNumber[nDice_] :=
Max@Table[RandomVariate[FairDie], {nDice}]
```

Mathematica example problem: Dice game involving value rolled.

Example from BinomialDistribution: Two players roll dice. If the total of both numbers is less than 10, the second player is paid 4 cents; otherwise the first player is paid 9 cents. Is the game fair?:

The first player wins 9 cents with probability

```
ℙPlayer1Wins = Module[{𝒟},
  𝒟 = DiscreteUniformDistribution[{1, 6}];
   Probability[roll1 + roll2 ≥ 10, roll1 ≈ 𝒟 && roll2 ≈ 𝒟]]
```

$\frac{1}{6}$

Then, the expected value of the game is

copyright © N T Gladd 2016

```
Module[{PayoffPlayer1 = 9, PayoffPlayer2 = 4},
 PayoffPlayer1 ℙPlayer1Wins - PayoffPlayer2 (1 - ℙPlayer1Wins)]
```

$-\frac{11}{6}$

which is negative for player 1.

Yet, what is the probability that player 1 is ahead after n games?

```
ℙPlayer1Ahead = Module[{p = ℙPlayer1Wins, PayoffPlayer1 = 9, PayoffPlayer2 = 4},
  Probability[PayoffPlayer1 x > PayoffPlayer2 (n - x), x ≈ BinomialDistribution[n, p]]]
```

$$\begin{cases} 6^{-n}\left(-5^n+6^n\right) & 1<n<\frac{13}{4} \\ 5^{-1+n}\times 6^{-n}\,n & n==1 \\ 5^{-1+n-\text{Floor}\left[\frac{4n}{13}\right]}\times 6^{-n}\,\text{Binomial}\left[n,1+\text{Floor}\left[\frac{4n}{13}\right]\right] & n\geq\frac{13}{4} \\ \quad \text{Hypergeometric2F1}\left[1,1-n+\text{Floor}\left[\frac{4n}{13}\right],2+\text{Floor}\left[\frac{4n}{13}\right],-\frac{1}{5}\right] & \\ 0 & \text{True} \end{cases}$$

```
DiscretePlot[ℙPlayer1Ahead, {n, 12},
 PlotStyle → Black, AxesLabel → {Stl["nGames"], Stl["ℙAhead"]},
 PlotLabel → Stl["Probability Player 1 is ahead after n games"]]
```



What is the expected value of this game after n rolls?

```
Module[{p = ℙPlayer1Wins, PayoffPlayer1 = 9, PayoffPlayer2 = 4, mean, variance},
 mean =
  Expectation[PayoffPlayer1 x - PayoffPlayer2 (n - x), x ≈ BinomialDistribution[n, p]];
 variance = Expectation[(x - mean)^2, x ≈ BinomialDistribution[n, p]];
 {mean, √variance }]
```

$\left\{-\frac{11\,n}{6}, \frac{1}{6}\sqrt{5\,n+144\,n^2}\right\}$

copyright © N T Gladd 2016

```
Module[{},
 Plot[{-11 n/6, -11 n/6 - 1/6 Sqrt[5 n + 144 n^2], -11 n/6 + 1/6 Sqrt[5 n + 144 n^2]}, {n, 1, 20}]]
```



What are the minimum and maximum expected values of a roll of four dice?

This is a *Mathematica* example problem under *Expectation/Applications*

```
Module[{𝒟min, 𝒟max, 𝔼min, 𝔼max, lab, g},
 𝒟min = OrderDistribution[{DiscreteUniformDistribution[{1, 6}], 4}, 1];
 𝔼min = N@Expectation[x, x ≈ 𝒟min];
 lab = Stl@StringForm["Expected minimum = ``", 𝔼min];
 g[1] = DiscretePlot[PDF[𝒟min, x], {x, 1, 6}, PlotRange → All, PlotStyle → Black,
   PlotLabel → lab, AxesLabel → {Stl["n"], Stl["ℙ[x = n]"]}];
 𝒟max = OrderDistribution[{DiscreteUniformDistribution[{1, 6}], 4}, 4];
 𝔼max = N@Expectation[x, x ≈ 𝒟max];
 lab = Stl@StringForm["Expected maximum = ``", 𝔼max];
 g[2] = DiscretePlot[PDF[𝒟max, x], {x, 1, 6}, PlotRange → All, PlotStyle → Black,
   PlotLabel → lab, AxesLabel → {Stl["n"], Stl["ℙ[x = n]"]}];
 Grid[{{g[1], g[2]}}]]
```



The following is a tricky followup question. What is the expected value of the three highest largest values. Note

$$\mathbb{E}(x1 + x2 + x3 + x4) = \mathbb{E}(x1) + \mathbb{E}(x2) + \mathbb{E}(x3) + \mathbb{E}(x4)$$

copyright © N T Gladd 2016

or

$$\mathbb{E}(x1 + x2 + x3 + x4) - \mathbb{E}(x1) = \mathbb{E}(x2) + \mathbb{E}(x3) + \mathbb{E}(x4)$$

```
Expectation[x1 + x2 + x3 + x4, {x1, x2, x3, x4} ≈
    ProductDistribution[{DiscreteUniformDistribution[{1, 6}], 4}]] - Expectation[
   x, x ≈ OrderDistribution[{DiscreteUniformDistribution[{1, 6}], 4}, 1]] // N
```

```
12.2446
```

What is the expected sum of 2 dice, x and y, conditional on the requirement that y ≤ 3?

First the unconditional expected value

```
Expectation[x + y,
 {x ≈ DiscreteUniformDistribution[{1, 6}], y ≈ DiscreteUniformDistribution[{1, 6}]}]
```

```
7
```

Then the conditional expected value.

```
Expectation[x + y  y ≥ 3,
 {x ≈ DiscreteUniformDistribution[{1, 6}], y ≈ DiscreteUniformDistribution[{1, 6}]}]
```

```
8
```

How many dice must be rolled to have a 95% probability of rolling a six?

The number of dice rolled to obtain a six is the same as the number of rolls required with a single die. The process is given by a Geometric distribution. As illustrated in the table, rolling 16 dice generates a probability of 95.49%.

copyright © N T Gladd 2016

```
Module[{p = 1/6, 𝒟, result, info},
 𝒟 = GeometricDistribution[p];

 result = Table[{x, CDF[𝒟, x] // N}, {x, 10, 20}];
 info = ({#[[1]], #[[2]]}) & /@ result;
 PrependTo[info, {"n dice rolled", "ℙ[6 appears]"}];
 LGrid[info, "Probability of rolling a six with n dice"] ]
```

Probability of rolling a six with n dice

n dice rolled	ℙ[6 appears]
10	0.865412
11	0.887843
12	0.906536
13	0.922113
14	0.935095
15	0.945912
16	0.954927
17	0.962439
18	0.968699
19	0.973916
20	0.978263

```
Module[{p = 1/6, 𝒟},
 𝒟 = GeometricDistribution[p];

 DiscretePlot[CDF[𝒟, x], {x, 0, 20}, PlotStyle → Black,
  ExtentSize → Right, AxesLabel → {Stl["nRolls"], Stl["ℙ"]},
  Epilog → {Red, Line[{{0, 0.95}, {20, 0.95}}] }] ]
```



This can be solved another way. The probability that there are no sixes in n rolls is

$$\mathbb{P}_{\text{no}-\text{sixes}} = \left(\frac{5}{6}\right)^n$$

so the probability that there is at least 1 six is

$$\mathbb{P}_{\text{at least one six}} = 1 - \left(\frac{5}{6}\right)^n$$

```
Clear@ℙ
```

```
w[1] = Solve[ℙ == 1 - (5 / 6)^n, n][[1, 1]]
```

$$n \to \text{ConditionalExpression}\left[\frac{2\,i\,\pi\,C[1]}{\text{Log}\left[\frac{6}{5}\right]} + \frac{\text{Log}\left[\frac{1}{1-\mathbb{P}}\right]}{\text{Log}\left[\frac{6}{5}\right]},\ C[1] \in \text{Integers}\right]$$

```
w[2] = w[1] /. C[1] → 0
```

$$n \to \frac{\text{Log}\left[\frac{1}{1-\mathbb{P}}\right]}{\text{Log}\left[\frac{6}{5}\right]}$$

```
w[3] = w[2] /. ℙ → 0.95
```

$n \to 16.431$

How many dice must be rolled to have a 95% probability of rolling a one and a two?

A simulation warmup.

copyright © N T Gladd 2016

```
Module[{nDiceMax = 10, nTrials = 10000, results, theory, lab, 𝒟, g},
 (* see below for theory *)
 theory = Table[{n, 1 - (2×5^n - 4^n)/6^n}, {n, 2, nDiceMax}] // N;
 results = Table[{nDice,
    Count[ProbabilityOneAndTwo[nDice, nTrials], 1] / nTrials // N},
   {nDice, 2, nDiceMax}];
 lab = Stl@StringForm["Probability that 1 and 2 appear in roll of
      n dice\nTheory (Black line)  Simulation (``)", nTrials];
 ListPlot[results, Epilog → {Black, Line[theory]} , PlotLabel → lab,
  AxesLabel → {Stl["nDice"], Stl["ℙ"]}]]
```



The problem is solved by counting. But, the exclusion principle must be used. Not having performed such calculations in a long time, I struggled a bit. The web site *http://math.stackexchange.com/questions/627848/probability-of-specific-result-of-dice-rolling-using-inclusion-exclusion-princip* discusses an analogous problem that made the solution problem clear.

Let A_1 and A_2 be the events that 1 and 2 never appear in a roll of n dice. The desired probability that both 1 and 2 appear is the probability of the event that the complement of $A_1 \cup A_2$ occur, Thus

$$\mathbb{P}\left[(A_1 \cup A_2)^C\right] = 1 - \mathbb{P}[(A_1 \cup A_2)] = 1 - (\mathbb{P}[A_1] + \mathbb{P}[A_2] - \mathbb{P}[A_1 \cap A_2])$$

The quantities on the rhs can be calculated.

$$\mathbb{P}[A_1] = \frac{5^n}{6^n} \quad \text{1 does not appear}$$

$$\mathbb{P}[A_2] = \frac{5^n}{6^n} \quad \text{2 does not appear}$$

$$\mathbb{P}[A_1 \cap A_2] = \frac{4^n}{6^n} \quad \text{neither 1 or 2 appear}$$

So

copyright © N T Gladd 2016

$$\mathbb{P}\left[(A_1 \cup A_2)^C\right] = 1 - \frac{2 \times 5^n - 4^n}{6^n}$$

If I didn't know the probabilities for A_1, etc. I could used *Mathematica* to calculate them. Consider the example of 2 dice. Let A_1 be the event that no 1 appears. Although this is easily determined to be $5^n/6^n$, I can use *Mathematica* to perform the counting.

```
ℙ["A₁"] = Length[Tuples[{2, 3, 4, 5, 6}, 2]] / 6^2
```

$$\frac{25}{36}$$

```
ℙ["A₂"] = Length[Tuples[{1, 3, 4, 5, 6}, 2]] / 6^2
```

$$\frac{25}{36}$$

The desired quantity is the probability of the complement of the event $A_1 \cup A_2$.

```
ℙ["A₁ ⋂ A₂"] = Length[Tuples[{3, 4, 5, 6}, 2]] / 6^2
```

$$\frac{4}{9}$$

The exclusion principle

$$A_1 \bigcup A_2 = A_1 + A_2 - A_1 \bigcap A_2$$

The intersection of A_1 and A_2 is the event that neither 1 or 2 appears

```
ℙ["A₁ ⋃ A₂"] = ℙ["A₁"] + ℙ["A₂"] - ℙ["A₁ ⋂ A₂"]
```

$$\frac{17}{18}$$

Then, the probability that at least a 1 and a 2 appear is the complement of the

```
ℙ["(A₁ ⋃ A₂)^C"] = 1 - ℙ["A₁ ⋃ A₂"]
```

$$\frac{1}{18}$$

This argument generalizes for n rolls to

$$\mathbb{P}\left[\text{"}(A_1 \bigcup A_2)^C\text{"}\right] = 1 - \left(\frac{5^n}{6^n} + \frac{5^n}{6^n} - \frac{4^n}{6^n}\right)$$

$$1 + \left(\frac{2}{3}\right)^n - \left(\frac{5}{3}\right)^n 2^{1-n}$$

copyright © N T Gladd 2016

The equation that determines how many dice must be rolled to be 95% confident that a 1 and a 2 will appear as

```
eqn =  1 - (5^n/6^n + 5^n/6^n - 4^n/6^n) == 0.95
```

$$1 + \left(\frac{2}{3}\right)^n - \left(\frac{5}{3}\right)^n 2^{1-n} == 0.95$$

This must be solved numerically.

```
FindRoot[eqn, {n, 10}][[1]]
```

```
n → 20.2025
```

So 21 dice must be rolled.

The case of how many rolls are required for a 1, a 2 and a 3 are required could be solved, in an analogous manner, by calculating the rhs of

```
ℙ["A1 ⋃ A2 ⋃ A3"] = ℙ["A1"] + ℙ["A2"] + ℙ["A3"] -
  ℙ["A1 ⋂ A2"] - ℙ["A1 ⋂ A3"] - ℙ["A2 ⋂ A3"] - ℙ["A1 ⋂ A2 ⋂ A3"]
```

Functions

```
Clear[ProbabilityOneAndTwo, ContainsOneAndTwo];
ProbabilityOneAndTwo[nDice_, nTrials_] :=
  Table[ContainsOneAndTwo[nDice], {nTrials}];

ContainsOneAndTwo[nMax_] :=
 Module[{listOfRolls, listOfCounts, firstOccurrence},
  listOfRolls = RandomVariate[DiscreteUniformDistribution[{1, 6}], nMax];
  (*Print[listOfRolls];*)
  {Count[listOfRolls, 1], Count[listOfRolls, 2]};
  If[(Count[listOfRolls, 1] ≥ 1) && (Count[listOfRolls, 2] ≥ 1), 1, 0]]
```

How many dice should be rolled to maximize the probability of rolling exactly one 6, exactly 2 6s, etc?

```
Clear[die];
die = DiscreteUniformDistribution[{1, 6}];
```

```
Probability[x == 6, x ≈ die] // N
```

```
0.166667
```

copyright © N T Gladd 2016

```
Probability[(x1 == 6 && x2 ≠ 6) ⋁ (x1 ≠ 6 && x2 == 6), {x1 ≈ die, x2 ≈ die}] // N
```

0.277778

In general, the probability of rolling 1 6 with the first of two die is $\frac{5}{6}\,\frac{1}{6}$ and with the second of two die is

```
2 5/36 // N
```

0.277778

With n dice

$$w[1] = n\left(\frac{1}{6}\right)\left(\frac{5}{6}\right)^{n-1}$$

$5^{-1+n} \times 6^{-n}\, n$

```
DiscretePlot[n (1/6) (5/6)^(n-1), {n, 1, 10}, PlotStyle → Black,
 AxesLabel → {Stl["nDice"], Stl["ℙ[one six]"]}]
```



Similarly, the probability of rolling exactly 2 sixes is

$$w[2] = \text{Binomial}[n, 2]\left(\frac{1}{6}\right)^{2}\left(\frac{5}{6}\right)^{n-2}$$

$2^{-1-n} \times 3^{-n} \times 5^{-2+n}\,(-1+n)\, n$

copyright © N T Gladd 2016

```
DiscretePlot[Binomial[n, 2] (1/6)^2 (5/6)^(n-2), {n, 2, 20},
 PlotStyle → Black, AxesLabel → {Stl["nDice"], Stl["ℙ[two sixes]"]}]
```



Show that the probability of throwing 14 is the same with 3 dice or 5 dice

```
Clear[die];
die = DiscreteUniformDistribution[{1, 6}];
```

```
Probability[x1 + x2 + x3 == 14, {x1 ≈ die, x2 ≈ die, x3 ≈ die}]
```

$\frac{5}{72}$

```
Probability[x1 + x2 + x3 + x4 + x5 == 14,
 {x1 ≈ die, x2 ≈ die, x3 ≈ die, x4 ≈ die, x5 ≈ die, x6 ≈ die}]
```

$\frac{5}{72}$

copyright © N T Gladd 2016

```
Module[{result3, result5},
 result3 = Table[{n, Probability[x1 + x2 + x3 == n, {x1 ≈ die, x2 ≈ die, x3 ≈ die}]},
   {n, 3, 18}];
 result5 = Table[{n, Probability[x1 + x2 + x3 + x5 + x5 == n,
     {x1 ≈ die, x2 ≈ die, x3 ≈ die, x4 ≈ die, x5 ≈ die}]}, {n, 3, 40}];
 ListPlot[{result3, result5}]]
```



6 dice are rolled. What is the distribution of the sum of their faces?

This problem is solved using the probability generating function pgf for a fair die. The pgf is a polynomial with coefficients equal to the probability of the outcome

$$\text{pgfDie} = \left(\frac{1}{6}x + \frac{1}{6}x^2 + \frac{1}{6}x^3 + \frac{1}{6}x^4 + \frac{1}{6}x^5 + \frac{1}{6}x^6\right)$$

$$\frac{x}{6} + \frac{x^2}{6} + \frac{x^3}{6} + \frac{x^4}{6} + \frac{x^5}{6} + \frac{x^6}{6}$$

The probability of the outcome of rolling 6 dice are the coefficients of the product (convolution) of the one die generating function

$$\text{distRoll6} = \text{pgfDie}^6 \text{ // Expand}$$

$$\frac{x^6}{46656} + \frac{x^7}{7776} + \frac{7x^8}{15552} + \frac{7x^9}{5832} + \frac{7x^{10}}{2592} + \frac{7x^{11}}{1296} + \frac{19x^{12}}{1944} + \frac{7x^{13}}{432} + \frac{43x^{14}}{1728} + \frac{833x^{15}}{23328} + \frac{749x^{16}}{15552} + \frac{119x^{17}}{1944} + \frac{3431x^{18}}{46656} + \frac{217x^{19}}{2592} + \frac{469x^{20}}{5184} + \frac{361x^{21}}{3888} + \frac{469x^{22}}{5184} + \frac{217x^{23}}{2592} + \frac{3431x^{24}}{46656} + \frac{119x^{25}}{1944} + \frac{749x^{26}}{15552} + \frac{833x^{27}}{23328} + \frac{43x^{28}}{1728} + \frac{7x^{29}}{432} + \frac{19x^{30}}{1944} + \frac{7x^{31}}{1296} + \frac{7x^{32}}{2592} + \frac{7x^{33}}{5832} + \frac{7x^{34}}{15552} + \frac{x^{35}}{7776} + \frac{x^{36}}{46656}$$

```
distRoll6 = Transpose[{Range[0, 36], CoefficientList[distRoll6, x]}];
```

copyright © N T Gladd 2016

```
ListPlot[distRoll6, PlotStyle → Black, AxesLabel → {Stl["n"], Stl["ℙ[n]]"]}]
```



The expectation is given by

```
Sum[(i - 1) distRoll6 [[i, 2]], {i, 1, 37}]
```

```
21
```

What is the probability of rolling the same number exactly three times with 5 dice.

This problem comes from
http://magoosh.com/gmat/2012/gmat-probability-difficult-dice-questions/

```
Clear[die];
die = DiscreteUniformDistribution[{1, 6}];
```

Specific cases can be calculated using

```
Probability[x1 == 1 && x2 == 1 && + x3 == 1 && x4 ≠ 1&& x5 ≠ 1,
 {x1 ≈ die, x2 ≈ die, x3 ≈ die, x4 ≈ die, x5 ≈ die}]
```

$$\frac{25}{7776}$$

The general solution involves combinatorics and logic. The probability of throwing the number 1 3 times with 3 dice is

$$\mathbb{P}^{(3)}_{111} = \left(\frac{1}{6}\right)^3$$

The probability of throwing any of the numbers 1, 2, ...,6 is

$$\mathbb{P}^{(3)}_{111,222,...,666} = 6\,\mathbb{P}_{111} = \left(\frac{1}{6}\right)^2$$

copyright © N T Gladd 2016

But there are 5 dice and Binomial[5, 3] = 10 ways of choosing the same three particular dice out of 5.

$$\mathbb{P}^{(5)}_{111,222,\ldots,666} = \binom{5}{3} \left(\frac{1}{6}\right)^2$$

Since exactly 3 occurrences of the same number is called for, the other two numbers must not be equal to the number that occurred 3 types. The probability that the remaining two numbers are different is

$$\mathbb{P}_{xx} = \left(\frac{5}{6}\right)^2$$

So the final answer is (with a = b = c, d ≠ a, e ≠ a)

$$\mathbb{P}^{(5)}_{abcde} = \binom{5}{3} \left(\frac{1}{6}\right)^2 \left(\frac{5}{6}\right)^2$$

```
Binomial[5, 3] (1/6)^2 (5/6)^2
```

$$\frac{125}{648}$$

```
Binomial[5, 3] (1/6)^2 (5/6)^2 // N
```

```
0.192901
```

copyright © N T Gladd 2016